

ФЕДЕРАЛЬНОЕ АГЕНТСТВО ПО ОБРАЗОВАНИЮ  
АСТРАХАНСКИЙ ГОСУДАРСТВЕННЫЙ УНИВЕРСИТЕТ

Ю.Ю. Тарасевич

Нахождение и визуализация автомодельных решений  
дифференциальных уравнений в частных  
производных средствами Maple

*Методические рекомендации по курсу  
«Линейные и нелинейные уравнения физики» для направления Физика  
и «Методы математической физики» для физико-математических  
и инженерных специальностей и направлений*

2010

УДК 517.957+004(075)

ББК

Нахождение и визуализация автомодельных решений дифференциальных уравнений в частных производных средствами Maple. Методические рекомендации / авт.-сост. Ю. Ю. Тарасевич. — Астрахань, 2010. — 23 с.

Рецензенты:

кандидат физико-математических наук, доцент Н. М. Ефремов;

кандидат физико-математических наук, доцент А. Г. Кушнер

В методических рекомендациях рассмотрено несколько частных, но практически значимых случаев нахождения и визуализации автомодельных решений и родственных им решений типа бегущей волны (линейное и нелинейное уравнения теплопроводности, уравнение Бюргерса, уравнение Кортевега-де Фриза) с помощью пакета Maple.

Методические рекомендации могут быть использованы в учебном процессе в качестве руководства к одному из разделов компьютерного практикума по курсу «Линейные и нелинейные уравнения физики» для направления «Физика» и «Методы математической физики» для физико-математических и инженерных специальностей и направлений.

# Содержание

<b>1. Автомодельные решения и бегущие волны .....</b>	<b>4</b>
<b>2. Примеры нахождения и визуализации автомодельных решений с помощью Maple.....</b>	<b>6</b>
2.1. Автомодельное решение линейного уравнения теплопроводности . . . . .	6
2.2. Автомодельное решение нелинейного уравнения теплопроводности . . . . .	10
2.3. Автомодельное решение уравнения Бюргерса . . . . .	14
2.4. Решение типа бегущей волны уравнения Бюргерса . . . . .	16
2.5. Решение типа бегущей волны уравнения Кортевега–де Фриза	18
<b>3. Упражнения.....</b>	<b>22</b>
<b>Рекомендуемая литература.....</b>	<b>23</b>

# 1. Автомодельные решения и бегущие волны

Для построения точных решений нелинейных уравнений математической физики разработан ряд методов, основанных на переходе к новым переменным (зависимым и независимым). При этом обычно ставится цель: найти новые переменные, число которых меньше, чем число исходных. Переход к таким переменным приводит к более простым уравнениям. В частности, поиск точных решений уравнений с частными производными с двумя независимыми переменными сводится к исследованию обыкновенных дифференциальных уравнений (или систем таких уравнений). Естественно, при указанной редукции решения обыкновенных дифференциальных уравнений дают не все решения исходного уравнения с частными производными, а лишь класс решений, обладающими некоторыми свойствами.

Наиболее простыми классами точных решений, которые описываются обыкновенными дифференциальными уравнениями, являются решения типа бегущей волны и автомодельные решения. Существование этих решений обычно (но не всегда) обусловлено инвариантностью рассматриваемых уравнений относительно сдвига и растяжения–сжатия.

Явление, развивающееся во времени, называется автомодельным, если распределения его характеристик в разные моменты времени получаются одно из другого преобразованием подобия [1]. Автомодельность позволяет во многих случаях свести задачу математической физики к решению обыкновенных дифференциальных уравнений, что существенно упрощает исследование. При помощи автомодельных решений исследователи пытаются увидеть характерные свойства новых явлений. Кроме того, автомодельные решения использовались как эталоны при оценке приближённых методов решения более сложных задач.

Автомодельность — это симметрия задачи, позволяющая скомпенсировать масштабные преобразования независимых переменных соответствующим растяжением решения [6].

$$u(x, t) = A(t) f(\xi), \quad \xi = \frac{x}{l(t)}.$$

В случае уравнений, зависящих от одной пространственной переменной, автомодельность означает выбор нового масштаба координаты  $l(t)$  и решения  $u(x, t)$  так, что в новых координатах решение является функцией одной переменной  $\xi$ .

Когда такое преобразование удаётся найти, задача сводится к обыкновенному дифференциальному уравнению. Иногда преобразование проще

всего найти с помощью анализа размерности физических величин, входящих в уравнение. Часто удобнее прямо анализировать преобразования растяжения неизвестных функций и независимых переменных и искать, какую группу преобразований допускает уравнение, т.е. какие растяжения оставляют уравнения неизменными.

В разнообразных проблемах математической физики важную роль играют инвариантные решения типа бегущих волн. Так называют решения, для которых распределения характеристик движения в разные моменты времени получаются одно из другого сдвигом, а не преобразованием подобия, как в случае автомодельных решений. Другими словами, всегда можно выбрать подвижную декартову систему координат так, что распределение характеристик движения типа бегущей волны в этой системе будет стационарно. К рассмотрению бегущих волн сводится исследование структуры фронта ударных волн в газодинамике и магнитной гидродинамике, уединённых и периодических волн в плазме и на поверхности тяжёлой жидкости, а так же многие другие задачи [1].

Решениями типа бегущей волны называются решения вида

$$u(x, t) = f(\xi), \quad \xi = x - Vt, \quad (1)$$

где величина  $V$  играет роль скорости распространения волны. Здесь  $x$  и  $t$  — пространственная и временная переменные,  $V$  — постоянная скорость распространения волны. Для таких решений распределения характеристик в разные моменты времени получаются простым сдвигом одного из другого.

Решения типа бегущих волн тесно связаны с автомодельными [4]. Действительно, если подставить  $x = \ln \chi$ ,  $t = \ln \tau$ ,  $\varphi(\ln \xi) = f(\xi)$  в (1), получим представление бегущей волны в автомодельном виде:

$$u(x, t) = \varphi \left( \ln \left( \frac{\chi}{\tau V} \right) \right) = f(\xi).$$

## 2. Примеры нахождения и визуализации автомодельных решений с помощью Maple

### 2.1. Автомодельное решение линейного уравнения теплопроводности

Найдем автомодельную подстановку для уравнения теплопроводности

$$\frac{\partial}{\partial t}u(x,t) = \frac{\partial^2}{\partial x^2}u(x,t) \quad (2)$$

с точечным начальным условием

$$u(x,0) = \delta(x), \quad (3)$$

анализируя растяжения неизвестной функции  $u(x,t)$  и переменных, при которых уравнение остаётся неизменным.

Потребуем, чтобы масштабирование переменных  $x' = \lambda x$ ,  $t' = \mu t$  и функции  $u'(x',t') = \nu u(x,t)$  не меняло вида уравнения (2) и начального условия (3).

Делая замену  $\frac{\partial u'(x',t')}{\partial t'} = \frac{\partial(\nu u(x,t))}{\partial(\mu t)}$  и  $\frac{\partial^2 u'(x',t')}{\partial (x')^2} = \frac{\partial^2(\nu u(x,t))}{\partial(\lambda x)^2}$ , видим, что

$$\frac{\nu}{\mu} \frac{\partial u(x,t)}{\partial t} = \frac{\nu}{\lambda^2} \frac{\partial^2 u(x,t)}{\partial x^2}.$$

Чтобы масштабирование не изменило уравнение (2), коэффициенты, появившиеся в левой и правой частях уравнения, должны быть одинаковыми. Таким образом, коэффициент  $\nu$  может быть произвольным, а

$$\mu = \lambda^2. \quad (4)$$

Будем искать автомодельную переменную в виде  $\xi = x^a t^b$ .

Автомодельная переменная не должна меняться при изменении масштаба переменных:

$$(\lambda x)^a (\mu t)^b = \lambda^a x^a \mu^b t^b = x^a t^b.$$

Следовательно,  $\lambda^a \mu^b = 1$ . С учётом (4)  $a = -2b$ .

Пусть  $b = -1/2$ , тогда  $a = 1$ , и автомодельная переменная имеет вид

$$\xi = \frac{x}{\sqrt{t}}.$$

Для определения коэффициента  $\nu$  воспользуемся одним из свойств  $\delta$ -функции

$$\delta(ax) = \frac{\delta(x)}{a}$$

и начальным условием (3).

Вспомним, что изменение масштабов независимых переменных может быть скомпенсировано преобразованием подобия автомодельного решения

$$\nu u(x, t) = u'(\lambda x, \mu t).$$

В начальный момент времени ( $t = 0$ )

$$\nu u(x, 0) = u'(\lambda x, 0),$$

но  $u(x, 0) = \delta(x)$ , а  $u'(\lambda x, 0) = \delta(\lambda x) = \frac{\delta(x)}{\lambda}$ , следовательно

$$\nu \delta(x) = \lambda^{-1} \delta(x).$$

Следовательно,  $\nu = \lambda^{-1}$ . Заметим, что переход к автомодельной переменной может быть записан как

$$u(x, t) = \nu^{-1} u\left(\frac{x}{\sqrt{t}}, \frac{t}{(\sqrt{t})^2}\right) = \nu^{-1} u\left(\frac{x}{\sqrt{t}}, 1\right),$$

то есть

$$\lambda = (\sqrt{t})^{-1},$$

$$\nu = \sqrt{t}.$$

Таким образом,

$$u(x, t) = \frac{f(\xi)}{\sqrt{t}}.$$

Найдем автомодельное решение с помощью пакета Maple<sup>1</sup>.

> restart;

Подключаем библиотеки.

---

<sup>1</sup>Соглашение об обозначениях. Здесь и далее команды Maple, вводимые пользователем, начинаются знаком > и набраны шрифтом, имитирующим печатную машинку. Формула, следующая непосредственно за командой и набранная курсивом, является результатом выполнения команды. Весь остальной текст и формулы являются комментариями.

```
> with(PDETools): with(ODETools): with(plots):
```

Задаём дифференциальное уравнение — одномерное уравнение теплопроводности.

```
> PDE :=diff(u(x,t),t) = diff(u(x,t),x$2);
```

$$PDE := \frac{\partial}{\partial t} u(x, t) = \frac{\partial^2}{\partial x^2} u(x, t)$$

Задаём автомодельную подстановку.

```
> tr:={t=tau,x=xi*tau^(1/2),u(x,t)=f(xi)/tau^(1/2)};
```

$$tr := \left\{ t = \tau, x = \xi \sqrt{\tau}, u(x, t) = \frac{f(\xi)}{\sqrt{\tau}} \right\}$$

Запишем уравнение теплопроводности (2) с использованием автомодельной подстановки.

```
> ODEA:= change(tr, PDE, simplify);
```

$$ODEA := -\frac{1}{2} \frac{f(\xi) + \xi \frac{d}{d\xi} f(\xi)}{\tau^{3/2}} = \frac{\frac{d^2}{d\xi^2} f(\xi)}{\tau^{3/2}}$$

Автомодельная подстановка привела к тому, что вместо дифференциального уравнения в частных производных мы получили обыкновенное дифференциальное уравнение. Знаменатель в виде  $\tau^{3/2}$  интереса не представляет. Избавимся от него.

```
> ODE1:=ODEA * tau^{3/2};
```

$$ODE1 := -\frac{1}{2} f(\xi) - \frac{1}{2} \xi \frac{d}{d\xi} f(\xi) = \frac{d^2}{d\xi^2} f(\xi)$$

Решаем с помощью функции `dsolve` полученное дифференциальное уравнение и упрощаем решение с помощью функции `simplify`.

```
> ODE2:=simplify(dsolve(ODE1));
```

$$ODE2 := f(\xi) = \left( \operatorname{erf}\left(\frac{1}{2} i\xi\right) \_C1 + \_C2 \right) e^{-\frac{1}{4} \xi^2}$$

Первую постоянную интегрирования найдем из условия, что на бесконечности решение и его первая производная должны обращаться в 0. Таким образом,  $\_C1 = 0$ .

```
> ODE3 := subs(_C1=0, ODE2);
```

$$ODE3 := f(\xi) = \_C2 e^{-\frac{1}{4} \xi^2}$$

Вторую постоянную интегрирования находим из предельного представления  $\delta$ -функции

$$\delta(x) = \lim_{k \rightarrow 0} \frac{e^{-\frac{x^2}{k}}}{\sqrt{k\pi}}$$

Поскольку  $u(x, 0) = \delta(x)$ , то

$$\lim_{t \rightarrow 0} \frac{f(\xi)}{\sqrt{t}} = \lim_{t \rightarrow 0} -C2 \frac{e^{-\frac{x^2}{4t}}}{\sqrt{t}} = \lim_{k \rightarrow 0} \frac{e^{-\frac{x^2}{k}}}{\sqrt{k\pi}}$$

и  $-C2 = \frac{1}{2\sqrt{\pi}}$ .

Подставим полученное значение постоянной интегрирования в решение.

```
> AMS:=subs(_C2=1/(2*Pi^(1/2)),ODE3);
```

$$AMS := f(\xi) = \frac{1}{2\sqrt{\pi}} e^{-\frac{1}{4}\xi^2}$$

Вернёмся к исходным переменным. Делаем обратную замену переменных с помощью функции `subs`.

```
> SOL:=subs({xi=x/sqrt(t),f(xi)=u(x,t)*sqrt(t)},AMS);
```

$$SOL := u(x, t) \sqrt{t} = \frac{1}{2} e^{-\frac{1}{4} \frac{x^2}{t}} (\sqrt{\pi})^{-1}$$

Решаем полученное уравнение с помощью функции `solve`.

```
> u := unapply(solve(SOL,u(x,t)), [x,t]);
```

$$u := (x, t) \mapsto \frac{1}{2} e^{-\frac{1}{4} \frac{x^2}{t}} \frac{1}{\sqrt{t}} \frac{1}{\sqrt{\pi}}$$


Таким образом, автомодельное решение уравнения теплопроводности с точечным начальным условием имеет вид

$$u(x, t) = \frac{1}{2\sqrt{\pi t}} e^{-\frac{x^2}{4t}}. \quad (5)$$

Автомодельное решение (5) описывает диффузионное расплывание и изображено на графике с помощью функции `plot`. Характерная ширина решения растёт как  $\sqrt{t}$ , максимальное значение уменьшается как  $\sqrt{t}$ , а площадь под графиком сохраняется.

```
> plot([u(x,0.025),u(x,0.05),u(x,0.1)],
> x=-1.5..1.5,u=-0.1..2,
> color=[red,blue,green],
> linestyle=[1,2,3],
> title="Автомодельное решение линейного уравнения
> теплопроводности");
```

### Автомодельное решение линейного уравнения теплопроводности



На компьютере можно посмотреть эволюцию решения с помощью функции `animate`. Для большей наглядности используем оси координат типа `boxed` и явно зададим диапазон по оси  $u$ .

```
> animate(plot,  
> [u(x,t),x=-1.5..1.5,u=-0.1..3,axes=boxed], t=0.01..0.1);
```

## 2.2. Автомодельное решение нелинейного уравнения теплопроводности

Найдем автомодельное решение нелинейного уравнения теплопроводности

$$\frac{\partial u(x,t)}{\partial t} = \frac{\partial}{\partial x} \left( u(x,t) \frac{\partial u(x,t)}{\partial x} \right) \quad (6)$$

с точечным начальным условием

$$u(x,0) = \delta(x). \quad (7)$$

*Задание.* Проведите самостоятельно вычисления, аналогичные приведённым в разделе 2.1, и найдите автомодельную подстановку.

Пусть  $b = -1/3$ , тогда  $a = 1$  и автомодельная переменная

$$\xi = \frac{x}{\sqrt[3]{t}}. \quad (8)$$

Переход к автомодельной переменной может быть записан как

$$u(x, t) = \frac{1}{\sqrt[3]{t}} f(\xi). \quad (9)$$

Найдем автомодельное решение нелинейного уравнения теплопроводности (6) с начальным условием (7), используя пакет Maple.

> restart;

Подключаем библиотеки.

> with(PDETools): with(ODETools): with(plots):

Задаём дифференциальное уравнение — нелинейное уравнение теплопроводности (6).

> PDE:=diff(u(x,t),t)=diff(u(x,t)\*(diff(u(x,t),x)),x);

$$PDE := \frac{\partial}{\partial t} u(x, t) = \left( \frac{\partial}{\partial x} u(x, t) \right)^2 + u(x, t) \frac{\partial^2}{\partial x^2} u(x, t)$$

Задаём автомодельную подстановку (8–9).

> tr := {t=tau, x=xi\*tau^(1/3), u(x,t)=f(xi)/tau^(1/3)};

$$tr := \left\{ t = \tau, x = \xi \sqrt[3]{\tau}, u(x, t) = \frac{f(\xi)}{\sqrt[3]{\tau}} \right\}$$

С помощью функции `dchange` запишем уравнение теплопроводности (6) с использованием автомодельной переменной.

> ODEA:=dchange(tr,PDE,simplify);

$$ODEA := -\frac{1}{3} \xi \frac{d}{d\xi} f(\xi) + f(\xi) = \frac{\left( \frac{d}{d\xi} f(\xi) \right)^2 + f(\xi) \frac{d^2}{d\xi^2} f(\xi)}{\tau^{4/3}}$$

Получили обыкновенное дифференциальное уравнение. Знаменатель  $\tau^{4/3}$  интереса не представляет, избавимся от него.

> ODE1 := ODEA\*tau^{4/3};

$$ODE1 := -\frac{1}{3} \xi \frac{d}{d\xi} f(\xi) - \frac{1}{3} f(\xi) = \left( \frac{d}{d\xi} f(\xi) \right)^2 + f(\xi) \frac{d^2}{d\xi^2} f(\xi)$$

Полученное обыкновенное дифференциальное уравнение можно один раз проинтегрировать.

> ODE2 := int(lhs(ODE1)-rhs(ODE1), xi);

$$ODE2 := -\frac{1}{3} \xi f(\xi) - f(\xi) \frac{d}{d\xi} f(\xi)$$

Пакет Maple не отображает постоянную интегрирования. При  $t \rightarrow 0$  и  $x \neq 0$ , решение должно исчезать, поэтому функция  $f(\xi)$  вместе со своей

производной должна обращаться в нуль при  $\xi \rightarrow \infty$ . Из этого условия находим, что постоянная интегрирования равна нулю.

Решаем с помощью функции `dsolve` полученное дифференциальное уравнение.

```
> ODE3 := dsolve(ODE2);
```

$$ODE3 := f(\xi) = 0, f(\xi) = -\frac{1}{6}\xi^2 + \_C1$$

Имеется два решения  $f(\xi) = 0$ , которое должно быть справедливо при больших по абсолютной величине  $\xi$ , и  $f(\xi) = -\frac{1}{6}\xi^2 + \_C1$ . По смыслу задачи  $f(\xi) \geq 0$ . Очевидно, что второе решение справедливо при  $\xi^2 < 6\_C1$ . Таким образом, решение склеивается из двух.

Найдем постоянную интегрирования из условия нормировки

$$\int_{-\infty}^{\infty} u(x, 0) dx = \int_{-\infty}^{\infty} f(\xi) d\xi = \int_{-\sqrt{6\_C1}}^{\sqrt{6\_C1}} f(\xi) d\xi = 1.$$

```
> f1 := unapply(rhs(ODE3[2]), xi)
```

$$f1 := \xi \mapsto -\frac{1}{6}\xi^2 + \_C1$$

```
> _C1:=solve(int(f1(xi),
```

```
> xi=-sqrt(6*_C1)..sqrt(6*_C1))=1, _C1)
```

$$\_C1 := \frac{1}{4} \sqrt[3]{6}$$

Подставим найденное значение постоянной интегрирования в решение и вернёмся к исходным переменным. Делаем обратную замену переменных с помощью функции `subs`.

```
> SOL:=subs({f(xi)=u(x,t)*t^(1/3),xi=x/t^(1/3)}, ODE3[2])
```

$$SOL := u(x, t) \sqrt[3]{t} = -\frac{1}{6} \frac{x^2}{t^{2/3}} + \frac{1}{4} \sqrt[3]{6}$$

Выражаем из полученного уравнения искомую функцию  $u(x, t)$ .

```
> AMS1 := solve(SOL, u(x, t))
```

$$AMS1 := \frac{1}{12} \frac{-2x^2 + 3\sqrt[3]{6}t^{2/3}}{t}$$

Сшиваем два решения.


```
> u:=unapply(piecewise(x^2/t^(2/3)<=(3/2)*6^(1/3),AMS1,0),[x,t])
```

$$u := (x, t) \mapsto \begin{cases} \frac{1}{12} \frac{-2x^2 + 3\sqrt[3]{6}t^{2/3}}{t} & \frac{x^2}{t^{2/3}} \leq 3/2 \sqrt[3]{6} \\ 0 & \text{otherwise} \end{cases}$$

На рисунке показано решение в четыре разных момента времени. В отличие от предыдущего примера, решение имеет резкий край — фронт тепловой волны. Расширение нагретой области в данном случае происходит по закону  $t^{1/3}$ .

```
> plot([u(x,0.01),u(x,0.02),u(x,0.1),u(x,0.5)],
> x=-2..2,y=-0.05..2.5,
> color=[red,blue,green,orange],
> linestyle=[1,5,3,2],
> thickness=3,
> labels=['x','u(x,t)'],
> title="Автомодельное решение нелинейного уравнения
> теплопроводности");
```

Автомодельное решение нелинейного уравнения теплопроводности



*Задание.* Самостоятельно постройте анимированный график, демонстрирующие эволюцию теплового фронта.

## 2.3. Автомоделное решение уравнения Бюргерса

Найдем автомоделное решение уравнения Бюргерса

$$\frac{\partial u}{\partial t} + u \frac{\partial u}{\partial x} = \frac{\partial^2 u}{\partial x^2} \quad (10)$$

с точечным начальным условием

$$u(x, 0) = \delta(x). \quad (11)$$

*Задание.* Найдите самостоятельно автомоделную подстановку.

Замена, оставляющая уравнение Бюргерса инвариантным, должна иметь вид  $u(x, t) = \lambda u(\lambda x, \lambda^2 t)$ . Если выбрать  $\lambda = \frac{1}{\sqrt{t}}$ , то

$$u(x, t) = \frac{f(\xi)}{\sqrt{t}}, \quad \xi = \frac{x}{\sqrt{t}}.$$

Найдем с помощью Maple автомоделное решение уравнения (10), удовлетворяющее начальному условию (11).

```
> restart;
```

Подключаем библиотеки.

```
> with(PDETools): with(ODETools): with(plots):
```

Задаём дифференциальное уравнение — уравнение Бюргерса (10).

```
> PDE := diff(u(x,t),t) + u(x,t)*(diff(u(x,t),x)) -
> (diff(u(x,t),x,x))=0;
```

$$PDE := \frac{\partial}{\partial t} u(x, t) + u(x, t) \frac{\partial}{\partial x} u(x, t) - \frac{\partial^2}{\partial x^2} u(x, t) = 0$$

Сделаем автомоделную подстановку в уравнение Бюргерса с помощью функции `dchange`.

```
> tr := {t=tau, x=xi*sqrt(tau), u(x,t)=f(xi)/tau^(1/2)};
```

$$tr := \left\{ t = \tau, x = \xi \sqrt{\tau}, u(x, t) = \frac{f(\xi)}{\sqrt{\tau}} \right\}$$

```
> ODEA := dchange(tr, PDE, simplify);
```

$$ODEA := \frac{1}{2} \frac{-\xi \frac{d}{d\xi} f(\xi) - f(\xi) + 2 f(\xi) \frac{d}{d\xi} f(\xi) - 2 \frac{d^2}{d\xi^2} f(\xi)}{\tau^{3/2}} = 0$$

Получили обыкновенное дифференциальное уравнение. Знаменатель интереса не представляет, от него можно избавиться.

```
> ODE1 := ODEA*tau^(3/2);
```

$$ODE1 := -\frac{1}{2} \xi \frac{d}{d\xi} f(\xi) - \frac{1}{2} f(\xi) + f(\xi) \frac{d}{d\xi} f(\xi) - \frac{d^2}{d\xi^2} f(\xi)$$

Полученное уравнение можно один раз проинтегрировать.

```
> ODE2 := int(ODE1,xi);
```

$$ODE2 := -\frac{1}{2} \xi f(\xi) + \frac{1}{2} (f(\xi))^2 - \frac{d}{d\xi} f(\xi)$$

Maple не отображает постоянную интегрирования. При  $t \rightarrow 0$  и  $x \neq 0$ , решение должно исчезать, поэтому функция  $f(\xi)$  вместе со своей производной должна обращаться в нуль при  $\xi \rightarrow \infty$ . Из этого условия находим, что постоянная интегрирования равна нулю.

Решим с помощью функции `dsolve` полученное дифференциальное уравнение.

```
> ODE3 := dsolve(ODE2);
```

$$ODE3 := f(\xi) = -2 \frac{e^{-\frac{1}{4}\xi^2}}{\sqrt{\pi} \operatorname{erf}\left(\frac{1}{2}\xi\right) - 2\_C1}$$

Вернемся к исходным переменным. Делаем обратную замену переменных с помощью функции `subs`.

```
> F1 := subs({xi=x/t^(1/2), f(xi)=u(x,t)*sqrt(t)}, PDE3);
```

$$F1 := u(x,t) \sqrt{t} = -2 e^{-1/4 \frac{x^2}{t}} \left( \sqrt{\pi} \operatorname{erf}\left(1/2 \frac{x}{\sqrt{t}}\right) - 2\_C1 \right)^{-1}$$

Таким образом, автомодельное решение уравнения Бюргерса имеет вид

```
> u := solve(F1, u(x, t));
```

$$u := -2 e^{-1/4 \frac{x^2}{t}} \left( \sqrt{t} \right)^{-1} \left( \sqrt{\pi} \operatorname{erf}\left(1/2 \frac{x}{\sqrt{t}}\right) - 2\_C1 \right)^{-1}$$

Постоянную интегрирования можно найти из начального условия, воспользовавшись предельным представлением функции Дирака и тем фактом, что  $\operatorname{erf}(+\infty) = 1$ .


```
> u:=unapply(simplify(subs(_C1=5/2*sqrt(Pi),u)),[x,t]);
```

$$u := (x,t) \mapsto -2 e^{-1/4 \frac{x^2}{t}} \frac{1}{\sqrt{t}} \frac{1}{\sqrt{\pi}} \left( \operatorname{erf}\left(1/2 \frac{x}{\sqrt{t}}\right) - 5 \right)^{-1}$$

На рисунке показано решение в четыре различных момента времени.

```
> plot([u(x,0.2),u(x,1),u(x,2),u(x,5)],
> x=-5..5,color=[red,blue,green,orange],linestyle=[1,2,3,5],
> title="Автомодельное решение уравнения Бюргерса");
```

Автомодельное решение уравнения Бюргерса



*Задание.* Самостоятельно постройте анимированный график, демонстрирующий эволюцию автомодельного решения уравнения Бюргерса.

## 2.4. Решение типа бегущей волны уравнения Бюргерса

Важным частным решением многих уравнений математической физики является решение типа бегущей волны  $u(x, t) = A(t)g(x - Vt)$ , где  $g$  — функция одной переменной.

Будем искать решение уравнения Бюргерса в виде бегущей ступеньки

$$f(\xi) = f(x - Vt),$$

где  $V$  — скорость волны. Причём

$$\lim_{x \rightarrow -\infty} u(x, t) = 1, \quad \lim_{x \rightarrow +\infty} u(x, t) = 0.$$

Найдем решение с помощью Maple.

> **restart**;

Подключаем библиотеки, которые могут потом понадобиться.

```
> with(ODETools): with(PDETools): with(plots):
```

Задаём дифференциальное уравнение — уравнение Бюргерса

```
> PDE := diff(u(x,t), t)+u(x,t)*(diff(u(x,t), x)) -
> mu*(diff(u(x,t), x, x)) = 0;
```

$$PDE := \frac{\partial}{\partial t}u(x,t) + u(x,t) \frac{\partial}{\partial x}u(x,t) - \mu \frac{\partial^2}{\partial x^2}u(x,t) = 0$$

Перейдем к новой переменной  $\xi = x - Vt$ , указав, что  $V$  является параметром.

```
> newvar := {t = tau, x = xi+V*tau, u(x,t) = v(xi)}
```

$$newvar := \{t = \tau, x = \xi + V\tau, u(x,t) = v(\xi)\}$$

```
> ODE := dchange(newvar, PDE, params = 'V')
```

$$ODE := -V \frac{d}{d\xi}v(\xi) + v(\xi) \frac{d}{d\xi}v(\xi) - \mu \frac{d^2}{d\xi^2}v(\xi) = 0$$

Получили обыкновенное дифференциальное уравнение, которое можно проинтегрировать с помощью функции `int`.

```
> ODE1 := int(lhs(ODE), xi);
```

$$ODE1 := -Vf(\xi) + \frac{1}{2}(f(\xi))^2 - \mu \frac{d}{d\xi}f(\xi)$$

Постоянную интегрирования Maple не отображает, она равна нулю, это видно из условия  $\lim_{x \rightarrow \infty} u(x,t) = 0$ , т.е. если  $\xi \rightarrow \infty$ , то  $f, f'$  и  $f'' \rightarrow 0$ .

Полученное с помощью интегрирования уравнение можно решить с помощью `dsolve`.

```
> SOL := dsolve(ODE1);
```

$$SOL := f(\xi) = 2V \left(1 + 2e^{\frac{V\xi}{\mu}} \_C1 V\right)^{-1}$$

Постоянную  $V$  находим из того что  $\lim_{x \rightarrow -\infty} u(x,t) = 1$ , т.е. если  $\xi \rightarrow -\infty$ , то  $f(\xi) = 1$ , откуда получаем  $V = 1/2$ . Если потребовать, чтобы  $f(0) = 1/2$ , то  $\_C1 = 1$ .

```
> subs({V = 1/2, _C1 = 1}, SOL);
```

$$f(\xi) = \left(1 + e^{1/2 \frac{\xi}{\mu}}\right)^{-1}$$

Удобно записать решение как функцию двух переменных: независимой переменной  $\xi$  и параметра  $\mu$ .

```
> f := unapply(rhs(SOL), [xi, mu]);
```

$$(\xi, \mu) \mapsto \left(1 + e^{1/2 \frac{\xi}{\mu}}\right)^{-1}$$


Представим автомодельное решение на графике, для этого воспользуемся функцией `plot`.

```

> plot([f(xi,0.01),f(xi,0.03),f(xi,0.1),f(xi,0.2)],xi=-1..1,
> color=[red,blue,green,orange],linestyle=[1,2,3,5],
> title=typeset("Фронт ударной
> волны при различных значениях параметра ",mu));

```

Фронт ударной волны уравнения Бюргера при различных значениях параметра  $\mu$



*Задание.* Самостоятельно постройте анимированный график, демонстрирующие эволюцию решения уравнения Бюргера в виде бегущей волны. Для этого следует вернуться к исходным переменным

```

> g := (x,t,mu) -> subs(xi = x-(1/2)*t, f(xi, mu))
g := (x,t,mu) ↦ f(x - 1/2 t, mu)

```

## 2.5. Решение типа бегущей волны уравнения Кортевега–де Фриза

Уравнение Кортевега–де Фриза

$$\frac{\partial}{\partial t}u(x,t) + 6u(x,t)\frac{\partial}{\partial x}u(x,t) + \frac{\partial^3}{\partial x^3}u(x,t) = 0$$

описывает волны в плазме, на мелкой воде и во многих других ситуациях, где имеется простейшая нелинейность и слабая дисперсия.

Найдем решение в виде бегущей волны  $u(x, t) = f(\xi)$ ,  $\xi = x - Vt$ , где  $V$  — скорость волны. При этом

$$\lim_{x \rightarrow -\infty} u(x, t) = 0, \lim_{x \rightarrow \infty} u(x, t) = 0.$$

Найдем решение с помощью Maple.

> restart;

Подключаем библиотеки

> with(PDETools): with(ODETools): with(plots):

Задаем уравнение Кортевега–де Фриза.

> PDE:=diff(u(x,t),t)+6\*u(x,t)\*diff(u(x,t),x)+diff(u(x,t),x\$3);

$$PDE := \frac{\partial}{\partial t} u(x, t) + 6 u(x, t) \frac{\partial}{\partial x} u(x, t) + \frac{\partial^3}{\partial x^3} u(x, t)$$

Задаём преобразования

> newvar := {t = tau, x = xi+V\*tau, u(x,t) = v(xi)}

$$newvar := \{t = \tau, x = \xi + V\tau, u(x, t) = v(\xi)\}$$

> ODE := dchange(newvar, PDE, params = 'V')

$$ODE := -V \frac{d}{d\xi} v(\xi) + 6 v(\xi) \frac{d}{d\xi} v(\xi) + \frac{d^3}{d\xi^3} v(\xi)$$

Проинтегрируем полученное обыкновенное дифференциальное уравнение с помощью функции int.

> ODE1:=int(ODE,xi);

$$ODE1 := -Vv(\xi) + 3(v(\xi))^2 + \frac{d^2}{d\xi^2}v(\xi)$$

Постоянная интегрирования, которую Maple не отображает, должна равняться 0, поскольку функция и её производные обращаются в 0 при  $\xi \rightarrow +\infty$ .

Заметим, что

$$\frac{d}{d\xi} \left( \left( \frac{d}{d\xi} v(\xi) \right)^2 \right) = 2 \left( \frac{d}{d\xi} v(\xi) \right) \frac{d^2}{d\xi^2} v(\xi).$$

Умножим полученное дифференциальное уравнение на  $2\frac{dv}{d\xi}$ . Тогда

$$\frac{d}{d\xi} \left( \left( \frac{d}{d\xi} v(\xi) \right)^2 \right) - 2 \left( \frac{d}{d\xi} v(\xi) \right) \left( Vv(\xi) - 3(v(\xi))^2 \right) = 0.$$

Умножаем на  $d\xi$  и интегрируем

$$\frac{1}{2} \left( \frac{d}{d\xi} v(\xi) \right)^2 + (v(\xi))^3 - \frac{1}{2} V (v(\xi))^2 + C_2 = 0.$$

Константу интегрирования выбираем равной нулю, т.к.  $f = f' = f'' = 0$ , при  $\xi \rightarrow +\infty$

Полученное уравнение ещё раз интегрируется

$$\ln \left( \frac{\sqrt{V} + \sqrt{V - 2v(\xi)}}{\sqrt{V} - \sqrt{V - 2v(\xi)}} \right) (\sqrt{V})^{-1} = \xi - C3$$

Константа  $C3$  имеет смысл начала отсчёта  $\xi$ . Если отсчитывать  $\xi$  от  $C3$ , получим

```
> pd :=
> ln((V^(1/2)+(V-2*v(xi))^(1/2))/(V^(1/2)-(V-2*v(xi))^(1/2)))/
V^(1/2)=xi;
```

$$pd := \ln \left( \frac{\sqrt{V} + \sqrt{V - 2v(\xi)}}{\sqrt{V} - \sqrt{V - 2v(\xi)}} \right) (\sqrt{V})^{-1} = \xi$$

```
> p := solve(pd, v(xi));
```

$$p := 2 \frac{Ve^{\xi\sqrt{V}}}{(1 + e^{\xi\sqrt{V}})^2}$$

Зададим функцию  $f$ , как функцию от двух переменных.


```
> f := unapply(p, [xi, V]);
```

$$f := (\xi, V) \mapsto 2 \frac{Ve^{\xi\sqrt{V}}}{(1 + e^{\xi\sqrt{V}})^2}$$

Полученное решение называется *простым солитоном* — это уединённая волна, у которой ширина, амплитуда и скорость задаются одним параметром  $V$ . Из рисунка видно, как с ростом скорости  $V$  уединённая волна становится более узкой и высокой.

```
> plot([f(xi,3),f(xi,6),f(xi,12),f(xi,24)],
> xi=-3..3,y=0.1..12,color=[red,blue,green,orange],
> linestyle=[1,2,3,5],title="Решение
> уравнения КдФ при различных значениях параметра V");
```

Решение уравнения КдФ при различных значениях параметра  $V$



*Задание.* Самостоятельно постройте анимированный график, демонстрирующий эволюцию решения уравнения Кортевега–де Фриза в виде бегущей волны. Для этого необходимо вернуться к исходным переменным

```
> g := (x, t, V) -> subs(xi = x-V*t, f(xi, V))
```

$$g := (x, t, V) \mapsto f(x - Vt, V)$$

### 3. Упражнения

1. Найти автомодельную переменную для уравнения Кортевега–де Фриза.
2. Найти автомодельное решение уравнения теплопроводности, если

$$u(x, 0) = \begin{cases} u_0, & \text{при } x > 0, \\ 0, & \text{при } x < 0. \end{cases}$$

3. Найти автомодельное решение модифицированного уравнения Бюргерса  $u_t + u^p u_x = u_{xx}$ , где  $p$  — натуральное число, если  $u(x, 0) = \delta(x)$ .
4. Найти решение в виде бегущей волны уравнения син–Гордона, удовлетворяющее условиям  $u(-\infty, 0) = 0$ ,  $u(+\infty, 0) = 2\pi$ .
5. Найти решение в виде уединённой бегущей волны уравнения нелинейной струны  $u_{tt} - u_{xx} + (u^2)_{xx} + u_{xxxx} = 0$ .
6. Найти решение типа ударной волны уравнения Бюргерса  $u_t + uu_x = \nu u_{xx}$ , удовлетворяющее условиям  $u(-\infty, 0) = u_1$ ,  $u(+\infty, 0) = u_2$ .
7. Найти автомодельное решение нелинейного уравнения теплопроводности  $u_t = (u^n u_x)_x$  с точечным начальным условием  $u(x, 0) = \delta(x)$ .

## Рекомендуемая литература

1. Баренблатт Г.И. Подобие, автомодельность, промежуточная асимптотика (теория и приложения к геофизической гидродинамике). Изд. 2-е. — Ленинград, Гидрометеиздат, 1982. — 256 с.
2. Баренблатт Г.И. Автомодельные явления — анализ размерностей и скейлинг / Пер. с англ.: Учебное пособие — Долгопрудный: Издательский Дом «Интеллект», 2009. — 216 с. ISBN 978-5-91559-017-4
3. Говорухин В.Н., Цибулин В.Г. Компьютер в математическом исследовании. Учебный курс. — СПб.: Питер, 2001. — 624 с.: ил. ISBN 5-272-00220-2
4. Полянин А.Д., Зайцев В.Ф., Журов А.И. Методы решения нелинейных уравнений математической физики и механики. — М.: Физматлит, 2005. ISBN 978-5-9221-0539-2
5. Данилов Ю.А. Лекции по нелинейной динамике. Элементарное введение. — М.: Либроком, 2010. 208 с. ISBN 978-5-397-00944-7 (Серия: Синергетика: от прошлого к будущему.)
6. Колоколов И.В., Кузнецов Е.А., Мильштейн А.И., Подивилов Е.В., Черных А.И., Шапиро Д.А., Шапиро Е.Г. Задачи по математическим методам физики. Изд. 3-е, испр. — М.: КомКнига, 2007. — 288 с. ISBN 978-5-484-00806-3 (ISBN 5-484-00806-9)